The "Infinitesimal Alexander Module"

Pensieve Header: Work on the "infinitesimal Alexander module" as in our (DBN and Zsuzsanna Dancso) paper "Finite Type Invariants of W-Knotted Objects: From Alexander to Kashiwara and Vergne" (http://www.math.-toronto.edu/~drorbn/papers/WKO/); continues pensieve://2009-06/.

```
<< KnotTheory`
```

```
Loading KnotTheory` version of April 20, 2009, 14:18:34.482.
Read more at http://katlas.org/wiki/KnotTheory.
```

```
A = Alexander[K = Knot[4, 1]][X]
```

KnotTheory::loading : Loading precomputed data in PD4Knots`.

$3-\frac{1}{X}-X$

```
G = GD @@ PD[K] /.
  X[i_, j_, k_, l_] :> If[PositiveQ[X[i, j, k, l]], Ar[l, i, +1], Ar[j, i, -1]]
```

```
GD[Ar[1, 4, 1], Ar[5, 8, 1], Ar[3, 6, -1], Ar[7, 2, -1]]
```

Drawing Arrow Diagrams

```
Draw[expr_] := expr /. gd_GD :> Draw[gd];
Draw[gd_GD] := Module[
  {n = Length[gd], h, k = 0},
  Graphics[{
    Line[{{0, 0}, {2 n + 1, 0}}],
    Table[Text[i, {i, -0.3}], {i, 2 n}],
    (List @@ gd) /. {
      Ar[i_, j_, s_] :> {
        h = Abs[i - j] / 2;
        BezierCurve[{
          {i, 0}, {i, h}, {(i + j) / 2, h}, {j, h}, {j, 0}
         }, SplineDegree → 2],
        Text[s * (++k), {(i + j) / 2, h - 0.3}],
        Line[{{j - 0.2, 0.4}, {j, 0}, {j + 0.2, 0.4}}]
       }
     }
   }
  ]
 ];
Draw[G]
```



Work in IAM

Conventions for red objects:

1. Legs start just to the right of the index; ar[0,7] means a red arrow starting to the right of position 0 (that is, to the left of everything) and ending to the right of position 7).
2. If two (red) indices are the same, the heads are to the right of the tails.
3. w1[] is the one-legged wheel object.
3. y[i,j,k] means "red Y with tails at i and j and head at k".

```
n = 2 Length[G]; range = Range[0, n];
Short[AllRedObjects = Flatten[{
    Outer[ar, range, range], Outer[y, range, range, range], w1[]
   }]]
```

```
{ar[0, 0], ar[0, 1], ar[0, 2], ar[0, 3], ar[0, 4],
 <<801>>, y[8, 8, 5], y[8, 8, 6], y[8, 8, 7], y[8, 8, 8], w1[]}
```

The relations associated with a red objects involve all the ways of "pulling one red leg one unit to the left". So d[i] means "a red leg at i minus a red leg at (i-1)":

```
ar[d[i_], j_] := ar[i, j] - ar[i - 1, j] + If[i - 1 == j, -w1[], 0];
ar[i_, d[j_]] := ar[i, j] - ar[i, j - 1]  + If[i == j,  w1[], 0];
y[d[i_], j_, k_] := y[i, j, k] - y[i - 1, j, k] + If[i - 1 == k, w1[], 0];
y[i_, j_, d[k_]] :=
  y[i, j, k] - y[i, j, k - 1] + If[i == k, -w1[], 0] + If[j == k, w1[], 0];
```

Now let's form all the red relations; starting with the anti-symmetry of y:

```
RedRelations = {};
RR[rel_RuleDelayed] := AppendTo[RedRelations, rel];
SetAttributes[RR, Listable];
RR[y[t1_, t2_, h_] * _. :> y[t1, t2, h] + y[t2, t1, h]];
```

The relations below match, one would hope, the picture at the bottom of http://www.math.toronto.edu/~drorbn/papers/WKO/.

```
RR@(                     Ar Tail                                      Ar Head
     ar      ar[i_, h_] Ar[i_, j_, s_] * _. :>           ar[j_, h_] Ar[i_, j_, s_] * _. :>
     Tail      ar[d[i], h]                                  ar[d[j], h] + (X^s - 1) y[i, j, h]
     ar     ar[t_, i_] Ar[i_, j_, s_] * _. :>          ar[t_, j_] Ar[i_, j_, s_] * _. :>
     Head    ar[t, d[i]] + (X^s - 1) y[i, t, j - 1]     ar[t, d[j]] - (X^s - 1) y[i, t, j - 1]  );

RR@(                       Ar Tail                                        Ar Head
     y         y[i_, t_, h_] Ar[i_, j_, s_] * _. :>            y[j_, t_, h_] Ar[i_, j_, s_] * _. :>
     Tail        y[d[i], t, h]                                   y[d[j], t, h] - (X^s - 1) y[i, j, h]
     y       y[t1_, t2_, i_] Ar[i_, j_, s_] * _. :>         y[t1_, t2_, j_] Ar[i_, j_, s_] * _. :>
     Head     y[t1, t2, d[i]] + (X^s - 1) y[t1, t2, j - 1]    y[t1, t2, d[j]] - (X^s - 1) y[t1, t2, j

RelationsIn[G_GD, red_] := ReplaceList[
   red * (Times @@ Select[G, (Intersection[List @@ #, List @@ red] =!= {}) &]),
   RedRelations
  ];
Short[AllRedRelations = Flatten[RelationsIn[G, #] & /@ AllRedObjects]]
```

```
{-ar[0, 0] + ar[0, 1] + (-1 + X) y[1, 0, 3],
 <<2167>>, -y[8, 8, 7] - (-1 + X) y[8, 8, 7] + y[8, 8, 8]}
```

```
rule = Dispatch[Thread[Rule[AllRedObjects, IdentityMatrix[Length[AllRedObjects]]]]];
Short[RedRules = Map[
   (
     p = First@Part[AllRedObjects, First@Position[#, 1, {1}]];
     p -> p - (#.AllRedObjects)
    ) &,
   DeleteCases[mat = Simplify[RowReduce[AllRedRelations /. rule]], {0 ...}]
  ]]
```

```
{ar[0, 0] -> ar[8, 8], ar[0, 1] -> ar[8, 8], <<805>>, y[8, 8, 7] -> 0, y[8, 8, 8] -> 0}
```

```
Simplify[RedRules[[Table[Random[Integer, {1, Length[RedRules]}], {10}]]]]
```

$$\left\{y[4,5,3]\to 0,\ y[2,2,5]\to 0,\ y[0,2,8]\to 0,\ y[5,5,5]\to 0,\ y[2,7,1]\to \frac{(-1+X)\,X\,w1[\,]}{1-3X+X^2},\ y[8,1,7]\to -\frac{w1[\,]}{X},\ y[5,4,0]\to 0,\ y[5,1,1]\to \frac{w1[\,]}{1-3X+X^2},\ y[5,0,7]\to 0,\ y[8,8,5]\to 0\right\}$$

```
lambda = Plus @@ G /. Ar[i_, j_, s_] :> s * ar[i, j]
```

```
ar[1, 4] - ar[3, 6] + ar[5, 8] - ar[7, 2]
```

```
deltaL = ar[0, 0] + w1[]; deltaR = ar[0, 0];
{
 slL = Plus @@ Cases[G, Ar[i_, j_, s_] /; j < i :> s],
 slR = Plus @@ Cases[G, Ar[i_, j_, s_] /; i < j :> s],
 SL = slL * deltaL + slR * deltaR
}
```

{-1, 1, -w1[]}

```
Simplify@{lambda - SL /. RedRules, X D[Log[A], X] * w1[]}
```

$$\left\{\frac{\text{w1}[\,] - X^2\,\text{w1}[\,]}{1 - 3X + X^2}, \frac{\left(-1 + X^2\right)\text{w1}[\,]}{1 - 3X + X^2}\right\}$$

There's still a sign issue above!

All G Matrices

Note. The product Sm.Dm here is called S in the paper.

```
Tij[Ar[ti_, hi_, si_], Ar[tj_, hj_, sj_]] := If[
   ti < hj < hi || hi < hj < ti,
   1, 0
  ];
Tm = Outer[Tij, List @@ G, List @@ G];
Sm = DiagonalMatrix[List @@ G /. Ar[_, _, s_] :> s];
Dm = DiagonalMatrix[List @@ G /. Ar[t_, h_, _] :> Sign[h - t]];
SDm = Sm.Dm;
Id = IdentityMatrix[n / 2];
SD1m = MatrixExp[-Log[X] SDm] - Id;
Bm = Tm.SD1m;
MatrixForm /@ {Tm, Sm, Dm, SDm, SD1m, Bm}
```

$$\left\{\begin{pmatrix} 0 & 0 & 0 & 1 \\ 0 & 0 & 1 & 0 \\ 1 & 0 & 0 & 0 \\ 1 & 0 & 1 & 0 \end{pmatrix}, \begin{pmatrix} 1 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 \\ 0 & 0 & -1 & 0 \\ 0 & 0 & 0 & -1 \end{pmatrix}, \begin{pmatrix} 1 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 \\ 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & -1 \end{pmatrix}, \begin{pmatrix} 1 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 \\ 0 & 0 & -1 & 0 \\ 0 & 0 & 0 & 1 \end{pmatrix},\right.$$

$$\left.\begin{pmatrix} -1 + \frac{1}{X} & 0 & 0 & 0 \\ 0 & -1 + \frac{1}{X} & 0 & 0 \\ 0 & 0 & -1 + X & 0 \\ 0 & 0 & 0 & -1 + \frac{1}{X} \end{pmatrix}, \begin{pmatrix} 0 & 0 & 0 & -1 + \frac{1}{X} \\ 0 & 0 & -1 + X & 0 \\ -1 + \frac{1}{X} & 0 & 0 & 0 \\ -1 + \frac{1}{X} & 0 & -1 + X & 0 \end{pmatrix}\right\}$$

```
Simplify@{lambda - SL /. RedRules, Tr[Bm.Inverse[Id - Bm].(Id + Tm).SDm] * w1[]}
```

$$\left\{\frac{\text{w1}[\,] - X^2\,\text{w1}[\,]}{1 - 3X + X^2}, \frac{\left(-1 + X^2\right)\text{w1}[\,]}{1 - 3X + X^2}\right\}$$

There's still a sign issue above!

The IAM Matrices

```
leftend[i_] := Min[G[[i, 1]], G[[i, 2]]];
rightend[i_] := Max[G[[i, 1]], G[[i, 2]]];
head[j_] := G[[j, 2]];
dir[j_] := Sign[head[j] - G[[j, 1]]];
eps = 0.2; (* For all practical purposes, this is "little" *)
lambij[i_, j_] := (
    ar[leftend[i] + eps, head[j] - eps * dir[j] / 2] -
     ar[rightend[i] - eps, head[j] - eps * dir[j] / 2]
   ) /. ar[a_, b_] :> (
     ar[Floor[a], Floor[b]] +
      If[Floor[a] == Floor[b] && a > b, w1[], 0]
    );
Lambda = Table[lambij[i, j], {i, n / 2}, {j, n / 2}];
Yij[i_, j_] := y[
    leftend[i] + eps, rightend[i] - eps, head[j] - eps * dir[j] / 2
   ] /. y[a_, b_, c_] :> (
     y[Floor[a], Floor[b], Floor[c]] +
      If[Floor[b] == Floor[c] && b > c, w1[], 0] +
      If[Floor[a] == Floor[c] && a > c, -w1[], 0]
    );
Ym = Table[Yij[i, j], {i, n / 2}, {j, n / 2}];
MatrixForm /@ {Lambda, Ym}
```

$$\left\{\begin{pmatrix} ar[1,3]-ar[3,3] & ar[1,7]-ar[3,7] & ar[1,5]-ar[3,5] & ar[1,2]-ar[3,2] \\ ar[5,3]-ar[7,3] & ar[5,7]-ar[7,7] & ar[5,5]-ar[7,5] & ar[5,2]-ar[7,2] \\ ar[3,3]-ar[5,3] & ar[3,7]-ar[5,7] & ar[3,5]-ar[5,5] & ar[3,2]-ar[5,2] \\ ar[2,3]-ar[6,3] & ar[2,7]-ar[6,7] & ar[2,5]-ar[6,5] & ar[2,2]-ar[6,2]+w1[] \end{pmatrix}, \begin{pmatrix} y[1,3,3] & y[1,3,7] & y[1,3,5] & y[1,3,2] \\ y[5,7,3] & y[5,7,7] & y[5,7,5] & y[5,7,2] \\ y[3,5,3] & y[3,5,7] & y[3,5,5] & y[3,5,2] \\ y[2,6,3] & y[2,6,7] & y[2,6,5] & -w1[]+y[2,6,2] \end{pmatrix}\right\}$$

Test 1

```
test1 = Simplify[{lambda - SL, Tr[Dm.Sm.Lambda]} /. RedRules]
```

$$\left\{\frac{w1[] - X^2\, w1[]}{1 - 3X + X^2}, \frac{w1[] - X^2\, w1[]}{1 - 3X + X^2}\right\}$$

```
{1, -1}.test1
```

0

Test 2

```
MatrixForm /@ (test2 = ExpandNumerator[Together[
    {Lambda, -Bm.Ym - Tm.(SD1m + Id)} /. RedRules /. w1[] → 1
   ]])
```

$$\left\{\begin{pmatrix} \frac{1-X}{1-3X+X^2} & 0 & \frac{X-X^2}{1-3X+X^2} & \frac{1}{1-3X+X^2} \\ \frac{-1+X}{1-3X+X^2} & 0 & \frac{-1+2X}{1-3X+X^2} & \frac{-1+2X-X^2}{X(1-3X+X^2)} \\ \frac{1}{1-3X+X^2} & 0 & \frac{1-2X+X^2}{1-3X+X^2} & \frac{1-X}{X(1-3X+X^2)} \\ \frac{X}{1-3X+X^2} & 0 & \frac{X^2}{1-3X+X^2} & \frac{1-X}{1-3X+X^2} \end{pmatrix}, \begin{pmatrix} \frac{1-X}{1-3X+X^2} & 0 & \frac{X-X^2}{1-3X+X^2} & \frac{1}{1-3X+X^2} \\ \frac{-1+X}{1-3X+X^2} & 0 & \frac{-1+2X}{1-3X+X^2} & \frac{-1+2X-X^2}{X(1-3X+X^2)} \\ \frac{1}{1-3X+X^2} & 0 & \frac{1-2X+X^2}{1-3X+X^2} & \frac{1-X}{X(1-3X+X^2)} \\ \frac{X}{1-3X+X^2} & 0 & \frac{X^2}{1-3X+X^2} & \frac{1-X}{1-3X+X^2} \end{pmatrix}\right\}$$

```
Simplify[{1, -1}.test2] // MatrixForm
```

$$\begin{pmatrix} 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \end{pmatrix}$$

Test 3

```
MatrixForm /@ (test3 = ExpandNumerator[Together[
    {Ym, Bm.Ym + Tm.(SD1m + Id)} /. RedRules /. w1[] → 1
   ]])
```

$$\left\{\begin{pmatrix} \frac{-1+X}{1-3X+X^2} & 0 & \frac{-X+X^2}{1-3X+X^2} & \frac{1}{-1+3X-X^2} \\ \frac{1-X}{1-3X+X^2} & 0 & \frac{1-2X}{1-3X+X^2} & \frac{1-2X+X^2}{X(1-3X+X^2)} \\ \frac{1}{-1+3X-X^2} & 0 & \frac{-1+2X-X^2}{1-3X+X^2} & \frac{-1+X}{X(1-3X+X^2)} \\ -\frac{X}{1-3X+X^2} & 0 & -\frac{X^2}{1-3X+X^2} & \frac{-1+X}{1-3X+X^2} \end{pmatrix}, \begin{pmatrix} \frac{-1+X}{1-3X+X^2} & 0 & \frac{-X+X^2}{1-3X+X^2} & -\frac{1}{1-3X+X^2} \\ \frac{1-X}{1-3X+X^2} & 0 & \frac{1-2X}{1-3X+X^2} & \frac{1-2X+X^2}{X(1-3X+X^2)} \\ -\frac{1}{1-3X+X^2} & 0 & \frac{-1+2X-X^2}{1-3X+X^2} & \frac{-1+X}{X(1-3X+X^2)} \\ -\frac{X}{1-3X+X^2} & 0 & -\frac{X^2}{1-3X+X^2} & \frac{-1+X}{1-3X+X^2} \end{pmatrix}\right\}$$

```
Simplify[{1, -1}.test3] // MatrixForm
```

$$\begin{pmatrix} 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \end{pmatrix}$$